Fourier Series using MAPLE

Consider Example 4 from the Notes:
f(t) = u(t) - u(t-2)   with T=4

```
> restart; alias(u=Heaviside):
> f:=t-> u(t)-u(t-2); T:=4:
```

$$f := t \rightarrow u(t) - u(t-2) \tag{1}$$

```
> plot(f(t),t=-8..8,thickness=2);
```



This plot will only produce one period of the function. To plot other parts of the function, recall f (t-T) will shift f (t) by an amount T to the right and f (t+T) will shift f (t) by an amount T to the left

```
> plot([f(t),f(t+T),f(t+2*T),f(t-T)],t=-8.00001..8,color=[red,blue,
  green,cyan],thickness=2);
```



To calculate the coefficients we need to evaluate the integrals:

```
> a0:=2/T*int(f(t),t=-T/2..T/2);
```

$$a0 := 1 \tag{2}$$

```
> an:=2/T*int(f(t)*cos(2*n*Pi*t/T),t=-T/2..T/2);
```

$$an := \frac{\sin(n\,\pi)}{n\,\pi} \tag{3}$$

```
> bn:=2/T*int(f(t)*sin(2*n*Pi*t/T),t=-T/2..T/2);
```

$$bn := \frac{1}{2} \; \frac{-2\cos(n\,\pi) + 2}{n\,\pi} \tag{4}$$

Notice that MAPLE makes no assumptions about n - we need to instruct MAPLE that n is an integer:

```
> assume(n,integer): an; bn:=simplify(bn);
```

$$0$$

$$bn := -\frac{(-1)^{n\sim} - 1}{n\sim\,\pi} \tag{5}$$

To view the Fourier Series as a summation:

```
> F:=a0/2 + Sum( an*cos(2*Pi*n*t/T) + bn*sin(2*Pi*n*t/T),n=1..
  infinity);
```

$$F := \frac{1}{2} + \sum_{n\sim=1}^{\infty} \left(-\frac{\left((-1)^{n\sim} - 1\right) \sin\left(\frac{1}{2}\, n\sim\,\pi\, t\right)}{n\sim\,\pi} \right) \tag{6}$$

Suppose we want to look at the first few terms in the series - say the terms up to N = 5

```
> N:=5:
> F5:=a0/2+sum(an*cos(2*Pi*n*t/T) + bn*sin(2*Pi*n*t/T),n=1..N);
```

$$F5 := \frac{1}{2} + \frac{2\sin\left(\frac{1}{2}\,\pi\, t\right)}{\pi} + \frac{2}{3}\;\frac{\sin\left(\frac{3}{2}\,\pi\, t\right)}{\pi} + \frac{2}{5}\;\frac{\sin\left(\frac{5}{2}\,\pi\, t\right)}{\pi} \tag{7}$$

```
> plot(F5,t=-8..8);
```



If we superimpose the original function for comparison;

```
> plot([F5,f(t)+f(t+T)+f(t+2*T)+f(t-T)],t=-8.0001..8,linestyle=
  [solid,dash]);
```


1
0.8
0.6
0.4
0.2
-8
-6
-4
-2
0
2
4
6
8
t