Eigenvalue Analysis

This document provides an illustration of the use of eigenvalue analysis to solve a set of differential equations. The set of equations considered is

$$\frac{\partial}{\partial t} x = 2x + 6y$$

$$\frac{\partial}{\partial t} y = -2x - 5y$$

To deal with this system we need to find the eigenvalues and eigenvectors of $\begin{bmatrix} 2 & 6 \\ -2 & -5 \end{bmatrix}$. To do this we call up the linear algebra package, set up the matrix and use the routines.

```
> restart; with(LinearAlgebra):
> A:= Matrix(1..2,1..2,[]):
> A[1,1]:=2: A[1,2]:=6:
> A[2,1]:=-2: A[2,2]:=-5:
> eval(A);
```

$$\begin{bmatrix} 2 & 6 \\ -2 & -5 \end{bmatrix} \tag{1}$$

```
> v:=Eigenvectors(A);
```

$$v := \begin{bmatrix} -1 \\ -2 \end{bmatrix}, \begin{bmatrix} -2 & -\frac{3}{2} \\ 1 & 1 \end{bmatrix} \tag{2}$$

The eigenvalues are the elements in the first column matrix and the eigenvectors are the columns of the second matrix.

```
> lam1:=v[1][1];
```

$$lam1 := -1 \tag{3}$$

```
> lam2:=v[1][2];
```

$$lam2 := -2 \tag{4}$$

We need to assign the second matrix of v to be P, as in the notes:

```
> P:= v[2];
```

$$P := \begin{bmatrix} -2 & -\frac{3}{2} \\ 1 & 1 \end{bmatrix} \tag{5}$$

If we solve the simplified system $\frac{\partial}{\partial t} z_i = \lambda_i z_i$ i= 1,2,3 and put the general solution into the vector q we obtain:

```
> z:=Vector([r*exp(lam1*t),s*exp(lam2*t)]);
```

$$z := \begin{bmatrix} r\,\mathrm{e}^{-t} \\ s\,\mathrm{e}^{-2t} \end{bmatrix} \tag{6}$$

since the eigenvalues are -1 and -2.

```
> xvec:= Multiply(P,z);
```

$$xvec := \begin{bmatrix} -2\,r\,\mathrm{e}^{-t} - \frac{3}{2}\,s\,\mathrm{e}^{-2t} \\ r\,\mathrm{e}^{-t} + s\,\mathrm{e}^{-2t} \end{bmatrix} \tag{7}$$

```
> xgs:=xvec[1]; ygs:=xvec[2];
```

$$xgs := -2\,r\,\mathrm{e}^{-t} - \frac{3}{2}\,s\,\mathrm{e}^{-2t}$$
$$ygs := r\,\mathrm{e}^{-t} + s\,\mathrm{e}^{-2t} \tag{8}$$

These expressions are the General Solutions of the system of differential equations. If we have any initial condtions we can now apply these and solve for the constants r and s. Suppose x(0)=1 and y(0) = 1

```
> eq1:= eval(subs(t=0,xgs)); eq2:=eval(subs(t=0,ygs));
```

$$eq1 := -2\,r - \frac{3}{2}\,s$$
$$eq2 := r + s \tag{9}$$

```
> ans:=solve({eq1=1,eq2=1},{r,s});
```

$$ans := \{r = -5, s = 6\} \tag{10}$$

```
> x:= subs(ans,xgs);    y:=subs(ans,ygs);
```

$$x := 10\,\mathrm{e}^{-t} - 9\,\mathrm{e}^{-2t}$$
$$y := -5\,\mathrm{e}^{-t} + 6\,\mathrm{e}^{-2t} \tag{11}$$

```
> plot([x,y],t=0..2,color=[red,blue]);
```



It is easy to perform other similar calculations by simply changing the elements of the matrix A. You can also deal with 3 by 3 systems by changing the specification of the arrays to 3 by 3 or 4 by 4 etc.